/**
 * Copyright (c) 2010 sonntagsfarben.de
 *
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the "Software"), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 *
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 *
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 */

package de.sonntagsfarben.puremvc.as3.mediator
{
	import de.sonntagsfarben.events.CoreEvent;

	import flash.events.Event;
	import flash.utils.Dictionary;

	import org.puremvc.as3.multicore.interfaces.INotification;
	import org.puremvc.as3.multicore.patterns.mediator.Mediator;

	/**
	 * @author Soenke Kluth <sonntagsfarben.de>
	 * @copyright (c) 2010 Soenke Kluth.
	 */
	public class CoreMediator extends Mediator
	{
		protected var notificationMap:Dictionary;


		public function CoreMediator(name:String = null, viewComponent:Object = null)
		{
			super(name, viewComponent);
			notificationMap = new Dictionary(true);
		}


		protected function sendEvent(event:Event, cancelEvent:Boolean = false):void
		{
			if (event is CoreEvent)
				sendNotification(event.type, (event as CoreEvent).data);
			else
				sendNotification(event.type, event, "Event");


			if (cancelEvent && event.cancelable)
			{
				event.stopImmediatePropagation();
			}
		}


		protected function mapNotificationHandler(notificationName:String, handler:Function):void
		{
			notificationMap[notificationName] = handler;
		}
		

		override public function listNotificationInterests():Array
		{
			var a:Array = new Array();

			for (var note:String in notificationMap)
			{
				a.push(note);
			}
			return a;
		}


		override public function handleNotification(notification:INotification):void
		{
			notificationMap[notification.getName()].apply(null, [notification]);
		}
		
		
		
		public override function onRemove():void
		{
			notificationMap = null;
			super.onRemove();
		}
	}
}
